

# Моделирование влияния преобразователей

## на сеть в среде системы Design — PSpice

Леонид Добрусин,  
Д. Т. Н.,  
академик АЭН РФ

antaniija@vei.ru

### 3. Графические символы и макромоделти типовых компонентов преобразовательной техники

#### 3.1. Общие положения

В настоящем разделе рассмотрена технология разработки графических символов типовых компонентов преобразовательной техники, которые отсутствуют в пакете штатных библиотек, поставляемых в комплекте с пакетом программ системы Design.

Главное внимание уделено процедурам разработки макромоделей и атрибутов графических символов. Методика создания чертежа графического символа дана в литературе [6, 7], поэтому в данной работе не рассматривается.

Материал настоящего и последующих разделов иллюстрируется примерами решения тестовых задач преобразовательной техники, начиная с простейших.

Представляется, что предлагаемая форма изложения материала поможет читателю лучше усвоить методику создания новых графических символов на базе макромоделей и успешно применять их при решении разных задач преобразовательной техники, возникающих при проектировании современных электротехнических комплексов и систем.

#### 3.2. Графический символ и макромоделть тиристора

Математические модели полупроводниковых приборов в системе Design описываются с помощью директив .MODEL или .SUBCKT. Директиве .MODEL соответствуют штатные модели компонентов, директиве .SUBCKT — макромоделти. Модели тиристоров представлены в виде макромоделей на языке программы PSpice, которые составлены на базе встроенных моделей диодов, транзисторов и других компонентов.

Встроенные модели и макромоделти полупроводниковых приборов учитывают большое количество параметров, поэтому их обработка требует значительных объемов вычислений. При моделировании сложных преобразователей, содержащих много полупроводниковых приборов, решение может не сходиться.

Для задачи анализа электромагнитных процессов в схемах силовых полупроводниковых преобразователей достаточно применять модели полупроводниковых приборов, которые отражают только их переключающие функции. Такие модели можно назвать логическими, так как состояние прибора

в данных моделях определяется логическими соотношениями между сигналами, приложенными к внешним узлам прибора.

Логические модели полупроводниковых приборов разрабатываются на языке программы PSpice путем составления макромоделей.

Логическая модель тиристора реализует естественный алгоритм его работы в качестве полууправляемого ключевого элемента силовой схемы преобразователя. Этот алгоритм формулируется следующим образом: «Если к аноду тиристора приложено положительное напряжение относительно катода и к управляющему электроду тиристора также приложено положительное напряжение относительно катода или значение прямого тока тиристора больше величины тока удержания, то тиристор находится в открытом состоянии; иначе — тиристор находится в закрытом состоянии».

Схема замещения, или функциональная модель тиристора (рис. 2а), соответствующая данному алгоритму, составлена на базе библиотечных компонентов системы Design и представляет собой три последовательно соединенных элемента: источник тока, управляемый приложенным к нему напряжением (GA); независимый источник напряжения (VA) и резистор (RG). Управляемый источник тока совместно с независимым источником напряжения имитируют силовую цепь тиристора, резистор — цепь управления. В целом данная цепь является моделью нелинейного резистора. Величина сопротивления этого резистора изменяется дискретно и может принимать одно из двух значений: значение сопротивления тиристора в открытом или закрытом состоянии.

Элементы схемы замещения тиристора выполняют следующие функции. Управляемый источник тока GA на языке программы PSpice используется для моделирования сопротивления силовой цепи тиристора между анодом (A) и катодом (K) в форме нелинейного резистора; независимый источник напряжения VA — для определения тока, протекающего через тиристор; резистор RG — для моделирования цепи управления тиристора между управляющим электродом (G) и катодом (K).

Стрелки на рис 2а соответствуют положительным направлениям напряжений, приложенных к элементам схемы замещения, и токов, протекающих через эти элементы:

- $V(A, pA)$ ,  $I(GA)$  — напряжение и ток источника тока GA;

- $V(pA, K)$ ,  $I(VA)$  — напряжение и ток источника напряжения GA;
- $V(G, K)$ ,  $I(RG)$  — напряжение и ток резистора RG.

Описание макромодели тиристора на языке программы Pspice, соответствующее схеме замещения на рис. 2а, имеет следующий вид (здесь и далее собственно текст описания выделен жирным шрифтом, строки комментариев содержат знак «\*»):

```
.SUBCKT SCR_m2_9 A G K
* .SUBCKT — директива начало описания макромодели;
* SCR_m2_9 — имя макромодели;
* A, G, K — маркировки узлов тиристора: анод, управляющий электрод, катод.
+ PARAMS: Ron=1e-3 Roff=1e+6 Iud=0.1
* PARAMS: — ключевое слово, после которого приводится перечень параметров макромодели;
* Ron — сопротивление тиристора в открытом состоянии, Ом;
* Roff — сопротивление тиристора в закрытом состоянии, Ом;
* Iud — ток удержания тиристора, А;
* «+» — знак продолжения строки.
.FUNC F1(x)={ IF ( x > 0 , x/Ron , x/Roff ) }
* .FUNC — описания функций, определяющих значение тока в силовой цепи тиристора между анодом (А) и катодом (К);
* алгоритм функции F1(x): если напряжение  $x > 0$ , то ток равен  $x/Ron$ , иначе —  $x/Roff$ .
.FUNC F2(x)={x/Roff}
* алгоритм функции F2(x): ток равен  $x/Roff$ .
RG G K 100
* предложение, описывающее включение резистора в схему по рис. 2а и его параметры:
* RG — имя резистора;
* G, K — маркировки узлов;
* 100 — значение сопротивления, Ом.
VA pA K 0
* предложение, описывающее включение независимого источника напряжения в схему по рис. 2а и его параметры:
* VA — имя источника;
* pA, K — маркировки узлов;
* 0 — значение напряжения, В.
GA A pA VALUE={ IF( (((V(G,K) > 0) & (V(A,K) > 0) )
+ (I(VA) > Iud) ) , F1(V(A,K)) , F2(V(A,K)) ) }
* предложение, описывающее включение источника тока в схему по рис. 2а и функцию, определяющую значения тока источника:
* GA — имя источника;
* A, pA — маркировки узлов;
* VALUE — ключевое слово, после которого приводится описание функции;
* алгоритм функции VALUE: если  $V(G,K) > 0$  и  $V(A,K) > 0$  или  $I(VA) > Iud$ , то значение тока источника GA вычисляется по формуле F1(x), иначе — по формуле F2(x), где x соответствует V(A,K);
* «+» — знак продолжения строки.
.ENDS SCR_m2_9
* .ENDS — директива окончания описания макромодели;
* SCR_m2_9 — имя макромодели.
На основе логической макромодели тиристора разработана новая модификация гра-
```




Рис. 2. Графический символ тиристора:  
а) функциональная схема макромодели  
б) чертеж графического символа

фического символа тиристора по методике редактирования графических символов программы Schematics.

Методика редактирования графических символов программы Schematics имеет два основных этапа:

1. Создание чертежа условного обозначения символа.
2. Создание атрибутов условного обозначения символа.

Для логической модели тиристора сохранен стандартный чертеж условного обозначения прибора и маркировки выводов (рис. 2б), что позволяет использовать данную модификацию символа тиристора при графическом вводе схемы наряду с моделями тиристоров из комплекта библиотек системы Design.

Список атрибутов логической модели тиристора имеет следующий вид:

```
PART=Thiristor_m3
* PART — шаблон имени символа компонента, под которым он заносится в библиотеку символов;
* Thiristor_m3 — имя символа компонента.
PKGREF=VS1
* PKGREF — шаблон позиционного обозначения компонента;
* 1 — позиционное обозначение компонента;
* VS1 — позиционное имя компонента.
```




Рис. 3. Чертеж модели однофазного двухполупериодного тиристорного выпрямителя, созданный в среде программы Schematics

REFDES=VS1

- \* REFDES — шаблон префикса условного обозначения компонента;
- \* VS — префикс условного обозначения компонента;
- \* VS1 — имя компонента.

```
TEMPLATE=X ^ @ REFDES %A %G %K @MODEL
```

- \* TEMPLATE — шаблон для назначения соответствий графических выводов компонента с их реальным физическим смыслом и задания параметров компонента;
- \* X — символ имени компонента, соответствующий макромодели;
- \* ^ — знак, который в процессе работы системы Design заменяется указанием полного пути доступа к компоненту;
- \* @ — знак, указывающий на то, что следующий за ним шаблон в процессе работы системы Design заменяется его именем;
- \* % — знак, предваряющий имя вывода компонента;
- \* MODEL — шаблон, обозначающий наличие описания модели или макромодели компонента.

MODEL=SCR\_m2\_9

- \* MODEL — шаблон имени модели или макромодели компонента;
- \* SCR\_m2\_9 — имя макромодели компонента.

**Пример 1. Моделирование схемы однофазного двухполупериодного тиристорного выпрямителя на основе логической макромодели тиристора.**

На данном примере рассмотрим основные процедуры моделирования схемы преобразователя в среде системы Design.

**Первая процедура** — графический ввод схемы преобразователя в среде программы Schematics и задание параметров компонентов. Результаты этого действия показаны на рис. 3 в виде окна программы Schematics с чертежом схемы преобразователя.

Поясним главные детали этого окна.

В верхней части экрана на черном (в оригинале — синем) фоне указано имя файла.

Ниже на сером фоне размещена группа позиций главного меню.




Рис. 4. Вид окна атрибутов источника синусоидального напряжения V2

Схема преобразователя, расположенная на поле экрана, состоит из компонентов, соединенных проводниками и шиной (физический аналог — многожильный кабель), которая на чертеже обозначена жирной линией.

Компоненты схемы последовательно введены на поле экрана по команде DRAW/GET NEW PART из библиотек графических символов, которые представляют собой файлы с расширением **.slb**. Графические символы V1, V2, V3, V4, L1 взяты из штатных библиотек системы Design; символы VS1, VS2, R1 — из библиотеки новых графических символов, созданной авторами [8, 9] и предназначенной для моделирования схем преобразовательной техники.

Графические символы V1, V2, V3, V4 находятся в библиотеке источников SOURCE.SLB под именами: V1, V2 — VSIN (источник синусоидального напряжения); V3, V4 — VPULSE (источник импульсного напряжения).

Графические символы V1, V2 моделируют источник синусоидальной э. д. с. с нулевым выводом, графические символы V3, V4 — систему управления преобразователя.

Параметры источников V1, V2 представлены на рис. 4, где показано окно атрибутов источника синусоидального напряжения V2:

- VAMP = 311 (В) — амплитуда напряжения;

- FREQ = 50 (Гц) — частота;
- VOFF = 0 (В) — постоянная составляющая;
- PHASE = 0 (град.) — начальная фаза (данный параметр находится в нижней части окна атрибутов, поэтому на рисунке отсутствует).

Параметры источника V2 аналогичны V1. Данные параметры можно редактировать путем выделения необходимой строки в окне атрибутов и последующей корректировки числа в графе Value.

Параметры источника V3 представлены на рис. 5, где показано окно атрибутов источника импульсного напряжения V3:

- V1 = 0 (В) — минимальное напряжение;
- V2 = 1v (В) — максимальное напряжение;
- TD = 2ms (мс) — задержка импульса;
- TR = 0.01us (мкс) — длительность переднего фронта;
- TF = 0.01us (мкс) — длительность заднего фронта;
- PW = 4ms (мс) — длительность импульса;
- PER = 20ms (мс) — период повторения импульса.

Параметры источника V4 аналогичны параметрам источника V3, кроме одного: TD = 12 мс, что обусловлено алгоритмом управления силовой схемой преобразователя.




Рис. 5. Вид окна атрибутов источника импульсного напряжения V3

Графический символ L1 находится в библиотеке аналоговых компонентов ANALOG.SLB под именем L (индуктивность). Значение индуктивности указано на рис. 3: «1 mH» (1 мГн). Данный параметр можно редактировать в окне атрибутов по аналогии с изложенным выше или в окне редактирования параметра, которое вызывается двойным щелчком мыши по параметру на поле экрана (рис. 6).

Графические символы VS1, VS2, R1 находятся в библиотеке новых графических символов USER\_SYMBOL.SLB под именами VS1, VS2 — «Thyristor\_m3»; R1 — R\_m1.

Вид окна атрибутов тиристора VS1 показан на рис. 7. В отличие от окон атрибутов, показанных на рис. 4–6, в данном окне можно редактировать только два атрибута: PKGREF — позиционное обозначение компонента на поле экрана и MODEL — имя макромодели. Значения численных параметров схемы замещения тиристора доступны для редактирования только в текстовом описании макромодели по команде меню EDIT/MODEL. При этом на поле экрана графический символ тиристора должен быть предварительно выделен.

Графический символ L1 отличается от штатного графического символа резистора, который находится в библиотеке аналоговых




Рис. 6. Вид окна редактирования параметра «1 mH»




Рис. 7. Вид окна атрибутов тиристора VS1




Рис. 8. Вид окна задания спецификации библиотек и текстовых файлов

компонентов ANALOG.SLB под именем R (резистор), только чертежом, так как чертеж резистора, принятый в системе Design, не соответствует ЕСКД. Новый чертеж резистора создан в режиме редактирования графических символов.

Параметры R1 можно редактировать по аналогии с изложенными выше способами редактирования параметров L1.

Некоторые проводники и шина имеют маркировки A1, G1, G2, K.

В общем случае, если все компоненты схемы находятся на поле и соединяются проводниками, последние можно не маркировать. Необходимость маркировок появляется в тех случаях, когда две части схемы соединяются с помощью шины или часть схемы описывается с помощью текстовых файлов. Данные приемы часто применяются при моделировании схемы в среде программы Schematics, чтобы не загромождать чертеж второстепенными деталями. Кроме того, необходимо маркировать те точки схемы, напряжение между которыми планируется подвергнуть гармоническому анализу.

**Вторая процедура** — задание спецификации библиотек и текстовых файлов, просмо-

триваемых при составлении списка соединенных компонентов схемы в среде программы Schematics. Результаты этого этапа показаны на рис. 8 в виде окна программы Schematics с наложенным окном редактирования спецификации библиотек и текстовых файлов (LIBRARY AND INCLUDE FILES).

Верхняя часть данного окна (File Name) предназначена для ввода или удаления необходимой информации в нижележащие части с помощью расположенной справа клавиатуры.

В части окна Library Files дана спецификация библиотек, просматриваемых при составлении списка соединенных компонентов схемы в данной задаче: nom.lib — спецификация штатных библиотек, две следующие строки указывают полный путь к библиотекам, созданным пользователем, то есть к библиотекам новых графических символов.

В части окна Include Files указан полный путь к текстовому файлу RC.txt, в котором на языке программы PSpice описана часть схемы, не показанная графически на чертеже схемы. Вид этого файла показан на рис. 9. В файле описана RC-цепь, подключенная между точками «0» и «K» схемы, показанной на рис. 3. Параметры RC-цепи: R = 500 Ом, C = 10 мкФ.

```
*****
.OPTIONS RELTOL=0.001 VNTOL=10mV ABSTOL=100mA
*****
C_1 0 0K 10u
R_2 0K K 500
*****
.end
```

Рис. 9. Вид текстового файла RC.txt

В первой строке файла дана директива установки параметров программы PSpice:

- .OPTIONS — имя директивы;
- RELTOL = 0.001 — допустимая относительная ошибка расчета напряжений и токов;
- VNTOL = 10mV — допустимая абсолютная ошибка расчета напряжений;
- ABSTOL = 100mA — допустимая абсолютная ошибка расчета токов.

**Третья процедура** — задание директив моделирования в среде программы Schematics, выполняется в два этапа. На первом этапе по команде ANALYSIS/SETUP на поле экрана вызывается панель директив моделирования Analysis Setup (рис. 10). На этой панели выбирается вид анализа, в нашем случае Transient — расчет переходных процессов во временной области и гармонический анализ. Затем щелчком мыши по кнопке «Transient» вызывается панель параметров моделирования в режиме Transient (рис. 11), где устанавливаются численные значения параметров режима моделирования.

Для нашего примера:

- параметры расчета переходных процессов:
  - Print Step: 20us (мкс) — шаг вывода результатов расчета;
  - Final Time: 30ms (мс) — конечное время расчета;
  - Step Ceiling: 40us (мкс) — максимальный шаг численного интегрирования дифференциальных уравнений;
- параметры гармонического анализа:
  - Center Frequency: 50 (Гц) — частота основной гармоники;
  - Number of harmonics: 20 — количество гармоник, учитываемых при анализе;




Рис. 10. Вид панели директив моделирования Analysis Setup




Рис. 11. Вид панели параметров моделирования в режиме Transient




Рис. 12. Вид панели установки параметров программы PROBE

– Output Vars.: V(A1,K) — координаты кривой напряжения, подлежащей гармоническому анализу; в данном случае — кривой напряжения, приложенного к тиристорам VS1.

**Четвертая процедура** — вызов из программы Schematics программы PSpice для расчета электромагнитных процессов в заданной схеме и автоматический вызов программы Probe для вывода результатов расчета в графической форме производится командой SIMULATE меню ANALYSIS при условии, что предварительно по команде

ANALYSIS/PROBE SETUP на панели установки параметров программы Probe (Probe Setup, рис. 12) установлен режим автоматического вызова программы Probe после окончания расчетов (Automatically Run Probe After Simulation).

Результаты моделирования схемы преобразователя, показанной на рис. 3, даны на рис. 13–14.

На рис. 13 показаны окна программ PSpice и Probe после окончания расчетов, наложенные на окно программы Schematics с моделируемой схемой.




Рис. 13. Результаты моделирования схемы однофазного двухполупериодного тиристорного выпрямителя

В нижней строке окна программы PSpice даны временные параметры расчета в секундах: Time step — длительность последнего шага численного интегрирования дифференциальных уравнений; Time — полное время расчета; End — конечное время расчета, заданное на панели параметров моделирования в режиме Transient (рис. 11) опцией Final Time. В процессе расчета значения параметров в строках Time step и Time меняются, что позволяет пользователю наблюдать процесс расчета и получать информацию для анализа при отладке модели.

В окне программы Probe показаны кривые напряжений на двух компонентах схемы, помеченных предварительно маркерами на схеме (рис. 3) по команде меню Markers/Mark Voltage Differential: напряжение на тиристоре VS1 (обозначение на рис. 13 — U(A1,K)) и напряжение управляющего импульса, приложенного от источника V3 к управляющему переходу тиристора VS1 (обозначение на рис. 13 — U(G1,K)).

На рис. 14 дан фрагмент файла test\_6\_2\_tiristora\_user\_i\_bus.out с результатами гармонического анализа напряжения V(A1,K).

Результаты моделирования схемы однофазного двухполупериодного тиристорного выпрямителя на основе логической макромоделли тиристора, представленные на рис. 13, 14, полностью подтверждают теорию преобразовательной техники, что свидетельствует об адекватном отображении физических свойств тиристора в рассмотренной логической макромоделли тиристора.

### 3.3. Графический символ и макромоделль системы управления

Макромоделль системы управления выполнена на базе встроенной модели PULSE импульсной функции, которая находится в штатной библиотеке SOURCE.SLB каталога математических моделей компонентов системы Design.

При моделировании источника импульса напряжения на основе импульсной функции PULSE задается следующий список параметров: начальное значение импульса (B); макси-




Рис. 14. Фрагмент файла TEST\_6\_2\_TIRISTORA\_USER\_I\_BUS.OUT с результатами гармонического анализа напряжения V(A1,K)




Рис. 15. Чертеж графического символа системы импульсно-фазового управления преобразователя, выполненного по трехфазной мостовой схеме




Рис. 16. Чертеж силовой схемы преобразователя, выполненного по трехфазной мостовой схеме, в комплекте с чертежом графического символа системы управления

мальное значение импульса (B); время задержки переднего фронта импульса относительно нуля системы отсчета (c); длительность переднего фронта импульса (c); длительность заднего фронта импульса (c); длительность плоской части импульса (c); период повторения импульсов (c). Указанный список параметров дает возможность создать макромодель системы управления для любой силовой схемы преобразователя.

В качестве примера рассмотрим макромодель системы импульсно-фазового управления преобразователя, выполненного по трехфазной мостовой схеме.

Чертеж силовой схемы преобразователя в комплекте с чертежом графического символа системы управления (рис. 15), выполненный в среде программы Schematics, показан на рис. 16. Собственно чертеж графического символа системы управления создан в среде программы Schematics по технологии редактирования графических символов. Диаграмма положения импульсов во временной области, обусловленная алгоритмом управления трехфазной мостовой схемой, показана на рис. 17.

На рис. 15–17 обозначено:

- F1...F6 — наименование каналов системы управления;

- CS1 — имя и позиционный номер графического символа системы управления;
- ALFHA=30 — угол управления, размерность — электрический градус;
- TETA\_0=30 — сдвиг точки отсчета угла управления A относительно нуля системы координат, размерность — электрический градус;
- T\_pulse=75 — длительность управляющего импульса, размерность — электрический градус.

В теории преобразовательной техники размерности временных параметров системы управления: длительность управляющего импульса, угол управления, сдвиг точки отсчета угла управления относительно нуля системы координат — принято задавать в электрических градусах.

Чтобы привести в соответствие размерности временных параметров импульсов, принятых в системе Design (c), с размерностями аналогичных параметров, принятых в силовой преобразовательной технике (электрические градусы), в текстовом описании макромодели применена функция, с помощью которой переводятся в секунды указанные параметры системы управления.

В качестве нуля системы координат принята первая точка пересечения положительной полуволны э. д. с. источника пи-




Рис. 17. Диаграмма положения импульсов во временной области

тания преобразователя Ea с осью абсцисс (рис. 17).

Текстовое описание макромодели системы импульсно-фазового управления трехфазной мостовой схемой преобразователя, соответствующей алгоритму управления трехфазной мостовой схемой (рис. 17), на языке программы PSpice имеет следующий вид:

```
.SUBCKT System_a1
+ 1n 1p 2n 2p 3n 3p 4n 4p 5n 5p 6n 6p
* .SUBCKT — директива начало описания макромодели;
* System_a1 — имя макромодели;
* 1n, 1p;...6n, 6p — маркировки узлов (выводов) системы управления, где p, n — обозначение выводов, которые должны соединяться с анодом и катодом тиристора соответственно; цифры перед буквенным обозначением вывода соответствуют нумерации тиристоров, принятой для трехфазной мостовой схемы в теории преобразовательной техники (рис. 16).
+ PARAMS: Tu=30 T0=0 A=30 Emin=-10
Emax=+15
* PARAMS: — ключевое слово, после которого приводится перечень параметров макромодели по умолчанию;
* Tu — длительность управляющего импульса; 30 — значение длительности уп-
```

равляющего импульса по умолчанию, размерность — электрический градус;

\* A — угол управления; 30 — значение угла управления по умолчанию, размерность — электрический градус;

\* T0 — сдвиг точки отсчета угла управления A относительно нуля системы координат; в качестве нуля системы координат принята первая точка пересечения положительной полуволны э. д. с. источника питания Ea с осью абсцисс (рис. 17);

\* Emin, Emax — начальное и максимальное значение импульса (B).

.FUNC F(x) (0.01/180) \*x

\* описание функции, с помощью которой производится пересчет размерности значений Tu, A, T0 в секунды.

VF1 1p 1n PULSE {Emin} {Emax} {F(T0+A)}

100u 100u

+ {F(Tu)} 20m

VF2 2p 2n PULSE {Emin} {Emax}

{F(T0+A+60)} 100u

+ 100u {F(Tu)} 20m

VF3 3p 3n PULSE {Emin} {Emax}

{F(T0+A+120)} 100u

+ 100u {F(Tu)} 20m

VF4 4p 4n PULSE {Emin} {Emax}

{F(T0+A+180)} 100u

+ 100u {F(Tu)} 20m

VF5 5p 5n PULSE {Emin} {Emax}

{F(T0+A+240)} 100u

+ 100u {F(Tu)} 20m

VF6 6p 6n PULSE {Emin} {Emax}

{F(T0+A+300)} 100u

+ 100u {F(Tu)} 20m

\* шесть предложений, описывающих включение импульсных источников напряжения в схему системы управления и их параметры: VF1... VF6 — имена источников; 1n, 1p;...6n, 6p — маркировки узлов; PULSE — имя встроенной модели импульсной функции; {Emin}, {Emax} — начальное и максимальное значение импульса; {F(T0+A+...)} — начало переднего фронта импульса (с); 100u, 100u — длительность переднего и заднего фронта импульса (мкс); {F(Tu)} — длительность плоской части импульса (с); 20m — период повторения импульсов (мс).

.ENDS System\_a1

\* .ENDS — директива окончания описания макромодели;

\* System\_a1 — имя макромодели.

Отметим, что ряд параметров в текстовом описании системы управления представлен в виде идентификаторов, помеченных фигурными скобками «{...}», что разрешает вводить в указанные тексты численные значения

данных параметров, заданных в директиве .SUBCKT после ключевого слова PARAMS.

Для редактирования численных значений главных параметров системы управления — длительности управляющего импульса, угла управления, сдвига точки отсчета угла управления относительно нуля системы координат — используется технология создания новых атрибутов в системе Design, позволяющая вводить указанные значения в режиме диалога с экрана монитора. При этом значения параметров Tu = 30 T0 = 0 A = 30, принятые по умолчанию в текстовом описании макромодели системы импульсно-фазового управления, заменяются соответственно значениями параметров T\_pulse TETA\_0 ALFHA, вводимыми с экрана монитора.

Рассмотренная макромодель системы импульсно-фазового управления применяется для моделирования преобразователей, которые питаются от сети частотой 50 Гц.

**Пример 2. Моделирование схемы тиристорного выпрямителя, выполненного по трехфазной мостовой схеме**

Цель данного примера — оценка корректности представления системы импульсно-фазового управления в виде графического символа для моделирования задач преобразовательной техники в среде системы Design.

Схема тестовой задачи в среде программы Schematics изображена на рис. 16.

Отметим, что при составлении схемы использован только один графический символ из каталога штатных библиотек системы Design — это графический символ L из библиотеки аналоговых компонентов ANALOG.SLB. С помощью этого символа моделируются все индуктивности: La, Lb, Lc, Ld.

Остальные графические символы: VS1...VS6, Rd, CS1, Ea1, Eb1, Ec1 — находятся в библиотеке новых графических символов USER\_SYMBOL.SLB под именами: VS1...VS6 — Thyristor\_m3; Rd — R\_m1; CS1 — System\_a1; Ea1 — Ea, Eb1 — Eb, Ec1 — Ec.

Остановимся на новых графических символах источников э. д. с.: Ea, Eb, Ec. Эти символы созданы путем редактирования чертежа и атрибутов штатного графического символа источника синусоидального напряжения, который находится в библиотеке источников SOURCE.SLB под именами VSIN.

Целесообразность создания данных символов продиктована тем, что трехфазные системы удобнее моделировать, имея графические символы фазных э. д. с.

При этом предусмотрено задание действующего значения э. д. с., как принято в практи-

ческих расчетах, в отличие от штатного графического символа, где задается амплитуда напряжения.

Вид панелей атрибутов графических символов источников синусоидальной э. д. с. Ea, Eb, Ec показан на рис. 18.

Поясним синтаксис шаблона TEMPLATE указанных атрибутов в той части, которая была отредактирована, начиная с позиции, следующей за параметром SIN:

- 0 — постоянная составляющая напряжения равна нулю;
- {SQRT(2)\*@E} — идентификатор пересчета действующего значения э. д. с. в амплитудное значение;
- 50 — частота (Гц);
- 0 — задержка относительно начала системы координат равна нулю;
- 0 — коэффициент затухания равен нулю;
- «0» (рис. 19а), «-120» (рис. 19б), «-240» (рис. 19в) — фазы источников синусоидальной э. д. с. Ea, Eb, Ec соответственно (град.).

Атрибут «E = 220» (B) задает действующее значение э. д. с. источников Ea, Eb, Ec. Поскольку новые графические символы этих источников базируются на штатной модели синусоиды (параметр SIN в шаблоне TEMPLATE), в которой требуется задавать амплитуду синусоиды, в шаблоне TEMPLATE предусмотрен идентификатор, реализующий процедуру расчета амплитуды по формуле  $E \times \sqrt{2}$ .

Все параметры компонентов схемы показаны на ее чертеже (рис. 16), в том числе и действующие значения источников э. д. с., в отличие от примера 1. При необходимости все параметры можно редактировать в режиме диалога с экрана монитора, как показано в примере 1 (рис. 6).

Покажем, каким образом в системе Design реализуется процедура вывода параметров графических символов на чертеж схемы с целью создания возможности для их редактирования в режиме диалога. Вначале по команде меню программы Schematics FILE/EDIT LIBRARY надо перейти из режима редактирования схемы в режим редактирования графических символов. Общий вид окна программы Schematics в указанном режиме дан на рис. 19. Затем по команде PART/COPY вызывается каталог библиотек графических символов, из которых выбирается и помещается в окно программы Schematics необходимый символ, в нашем примере — символ Ea из библиотеки USER\_SYMBOL.SLB (см. левую часть рис. 19). Далее по команде PART/ATTRIBUTES вызывается панель редактирования атрибутов (см. правую часть рис. 19),




Рис. 18. Вид панелей атрибутов графических символов источников синусоидальной э. д. с. а) Ea; б) Eb; в) Ec

где выделяется необходимый атрибут и в окне Display отмечаются опции Value (значение) и Name (имя), как показано на рис. 19 для атрибута E=220.

Результаты моделирования схемы тиристорного выпрямителя, выполненного по трехфазной мостовой схеме, предоставлены на рис. 20, где обозначено: I(La) — кривая тока в индуктивности La; U(A1,DK) — напряжение на тиристоре VS1. Форма и параметры данных кривых соответствуют их теоретическим аналогам, известным из теории трехфазной мостовой схемы, что свидетельствует о корректности представления системы импульсно-фазового управления в виде рассмотренного графического символа для моделирования задач преобразовательной техники в среде системы Design.

### Литература

1. Кудрявый В. В. Перспективы развития электроэнергетики России // Электро. 2003. № 1.
2. Карташов И. И., Пономаренко И. С., Тульский В. Н., Шамонов Р. Г., Масленников Г. К., Васильев В. В. Качество электрической энергии в муниципальных сетях Московской области // Промышленная энергетика. 2002. № 8.
3. Федеральный закон РФ «О техническом регулировании» № 184-ФЗ от 27 декабря 2002 года // Российская газета. 31 декабря 2002. № 245 (3113).
4. Федеральный закон РФ «Об электроэнергетике» № 35-ФЗ от 26 марта 2003 года // Российская газета. 1 апреля 2003. № 60 (3174).
5. Правила по сертификации. Система сертификации ГОСТ Р. О внесении изменений и дополнений в правила проведения сертификации электрооборудования. Утверждены Постановлением Госстандарта России № 1 от 3 января 2001 года.
6. Разевиг В. Д. Система схемотехнического моделирования и проектирования печатных плат Design Center (PSPICE). М.: СК Пресс, 1996.
7. Разевиг В. Д. Система сквозного проектирования электронных устройств DesignLab 8.0 М.: Солон, 1999.
8. Добрусин Л. А., Шитов В. А. Модели трансформаторов для анализа силовых полупроводниковых преобразователей в системе Design. Электротехника 2010. VI симпозиум. Сборник докладов, том III. М.: ВЭИ-ТРАВЭК, 2001.
9. Добрусин Л. А., Шитов В. А. Логическая модель тиристора для анализа силовых полупроводниковых преобразователей в системе Design. Электротехника 2010. VI симпозиум. Сборник докладов, том III. М.: ВЭИ-ТРАВЭК, 2001.
10. Добрусин Л. А., Шитов В. А. Макромодель системы управления преобразователем. Электротехника 2010. VII симпозиум. Сборник докладов, том I. М.: ВЭИ-ТРАВЭК, 2003.
11. Добрусин Л. А., Шитов В. А. Макромоделирование сложных преобразователей // Электро. 2002. № 6.




Рис. 19. Вид окна программы Schematics в режиме редактирования атрибутов графического символа Ea




Рис. 20. Результаты моделирования схемы тиристорного выпрямителя, выполненного по трехфазной мостовой схеме

12. Добрусин Л. А. Многомостовые преобразователи в электроэнергетике XXI века. Электротехника 2010. V симпозиум. Сборник докладов, том II. М.: ВЭИ-ТРАВЭК, 1999.
13. Добрусин Л. А. Фильтрокомпенсирующие устройства для преобразовательной техники. М.: НТФ «Энергопрогресс», 2003.
14. Добрусин Л. А. Автоматизация расчета гармоник в электрических сетях, питающих преобразователи // Промышленная энергетика. 2003. № 4.
15. Kimbark E.W. Direct current transmission. Wiley interscience. 1971.
16. Добрусин Л. А. Универсальная модель для исследования качества электроэнергии в цепях с преобразователями. Электротехника 2010. VII симпозиум. Сборник докладов, том I. М.: ВЭИ-ТРАВЭК, 2003.
17. Добрусин Л. А. Методология и библиотека моделей для анализа влияния преобразователей на качество электроэнергии // Электро. 2003. № 5.
18. Добрусин Л. А. Выбор способа ограничения добротности силового индуктивно-емкостного фильтра // Электротехника. 1984. № 5.
19. Добрусин Л. А. Особенности моделирования преобразователей в среде системы Design // Электротехника 2010. VII симпозиум. Сборник докладов, том I. М.: ВЭИ-ТРАВЭК, 2003.
20. Добрусин Л. А. Широкополосные фильтрокомпенсирующие устройства для тиристорных преобразователей // Электричество. 1985. № 4.
21. Добрусин Л. А. Автоматизация расчета фильтрокомпенсирующих устройств для электрических сетей, питающих преобразователи // Промышленная энергетика. 2004. № 5.